cd /disk/agedisk3/medicare.work/poterba-DUA52260/jetson-dua52260/outliers/synth/100pct/robustness/feout
use types_panel_feout.dta, clear  
keep if group == 999

merge 1:1 dt using  synth_fit.dta

local treatment_period = mofd(mdy(11, 4, 2002))
disp `treatment_period'
	* 514

** Produce difference in series
gen deterrence = fit1 - out if dt >= `treatment_period'

** Time discount the difference at 10% 
local r = 1.1^(1/12)

gen deterrence_disc = deterrence*1/(`r'^(dt-`treatment_period'))
replace deterrence_disc = . if dt-`treatment_period' >= 60
count if !mi(deterrence_disc)
assert(r(N)==60)

*** Compute total
log using deterrence_feout.log, replace
preserve
collapse (sum) deterrence_disc
qui summarize deterrence_disc
local deterrence = r(min)
disp `deterrence'
restore
log close

*** Truncate graph at 5 years 
drop if dt  - `treatment_period' >= 60
summarize dt 
local minperiod = r(min)
local maxperiod = r(max)

*** Settlement date: June 29, 2006 
local settlement = mofd(mdy(6,29,2006))

format dt %tmMon_YY
twoway (scatter pmt_amt dt, mcolor(gs5)) (line out fit1 dt, lwidth(thick thick) lcolor(navy maroon)), xline(`treatment_period') xline(`settlement', lcolor(black)) legend(order(2 "Outliers" 3 "Synthetic Control") width(80)) xscale(range(`minperiod' `maxperiod')) xlabel(468 492 `treatment_period' 540 564) xtitle("Month") ytitle("$/Month") graphregion(color(white)) yscale(range(-700000000 400000000)) ylabel(-600000000 "-$600 Mil" -300000000 "-$300 Mil" 0 "$0" 300000000 "$300 Mil")

graph save "OutliersOverlay_feout.gph", replace


